

# NAG Fortran Library Routine Document

## F03ACF

**Note:** before using this routine, please read the Users' Note for your implementation to check the interpretation of *bold italicised* terms and other implementation-dependent details.

### 1 Purpose

F03ACF calculates the determinant of a real symmetric positive-definite band matrix using a Cholesky factorization.

### 2 Specification

```
SUBROUTINE F03ACF(A, IA, N, M, DET, RL, IL, M1, IFAIL)
INTEGER          IA, N, M, IL, M1, IFAIL
real           A(IA,M1), DET, RL(IL,M1)
```

### 3 Description

The determinant of  $A$  is calculated using the Cholesky factorization  $A = LL^T$ , where  $L$  is a lower triangular band matrix. The determinant of  $A$  is the product of the squares of the diagonal elements of  $L$ .

### 4 References

Wilkinson J H and Reinsch C (1971) *Handbook for Automatic Computation II, Linear Algebra* Springer-Verlag

### 5 Parameters

1:  $A(IA,M1)$  – *real* array *Input*

*On entry:* the lower triangle of the  $n$  by  $n$  positive-definite symmetric band matrix  $A$ , with the diagonal of the matrix stored in the  $(m + 1)$ th column of the array, and the  $m$  sub-diagonals within the band stored in the first  $m$  columns of the array. Each row of the matrix is stored in the corresponding row of the array. For example, if  $n = 5$  and  $m = 2$ , the storage scheme is

$$\begin{pmatrix} * & * & a_{11} \\ * & a_{21} & a_{22} \\ a_{31} & a_{32} & a_{33} \\ a_{42} & a_{43} & a_{44} \\ a_{53} & a_{54} & a_{55} \end{pmatrix}.$$

The elements in the top left corner of the array are not used. The following code may be used to assign elements within the band of the lower triangle of the matrix to the correct elements of the array:

```
DO 20 I = 1, N
  DO 10 J = MAX(1,I-M), I
    A(I,J-I+M+1) = matrix(I,J)
  10 CONTINUE
  20 CONTINUE
```

2:  $IA$  – INTEGER *Input*

*On entry:* the first dimension of the array  $A$  as declared in the (sub)program from which F03ACF is called.

*Constraint:*  $IA \geq N$ .

- 3: N – INTEGER *Input*  
*On entry:*  $n$ , the order of the matrix  $A$ .
- 4: M – INTEGER *Input*  
*On entry:*  $m$ , the number of sub-diagonals within the band of  $A$ .
- 5: DET – *real* *Output*  
*On exit:* the determinant of  $A$ .
- 6: RL(IL,M1) – *real* array *Output*  
*On exit:* the lower triangular matrix  $L$ , stored in the same way as  $A$ , except that in place of the diagonal elements, their reciprocals are stored.
- 7: IL – INTEGER *Input*  
*On entry:* the first dimension of the array RL as declared in the (sub)program from which F03ACF is called.  
*Constraint:*  $IL \geq N$ .
- 8: M1 – INTEGER *Input*  
*On entry:* the value  $m + 1$ .
- 9: IFAIL – INTEGER *Input/Output*  
*On entry:* IFAIL must be set to 0,  $-1$  or 1. Users who are unfamiliar with this parameter should refer to Chapter P01 for details.  
*On exit:* IFAIL = 0 unless the routine detects an error (see Section 6).  
 For environments where it might be inappropriate to halt program execution when an error is detected, the value  $-1$  or 1 is recommended. If the output of error messages is undesirable, then the value 1 is recommended. Otherwise, for users not familiar with this parameter the recommended value is 0. **When the value  $-1$  or 1 is used it is essential to test the value of IFAIL on exit.**

## 6 Error Indicators and Warnings

If on entry IFAIL = 0 or  $-1$ , explanatory error messages are output on the current error message unit (as defined by X04AAF).

Errors or warnings detected by the routine:

IFAIL = 1

The matrix  $A$  is not positive-definite, possibly due to rounding errors.

IFAIL = 2

Overflow. The value of the determinant is too large to be held in the computer.

IFAIL = 3

Underflow. The value of the determinant is too small to be held in the computer.

## 7 Accuracy

The accuracy of the determinant depends on the conditioning of the original matrix. For a detailed error analysis see page 54 of Wilkinson and Reinsch (1971).

## 8 Further Comments

The time taken by the routine is approximately proportional to  $n(m+1)^2$ .

This routine should only be used when  $m \ll n$  since as  $m$  approaches  $n$ , it becomes less efficient to take advantage of the band form.

## 9 Example

To calculate the determinant of the real symmetric positive-definite band matrix

$$\begin{pmatrix} 5 & -4 & 1 & & & & \\ -4 & 6 & -4 & 1 & & & \\ 1 & -4 & 6 & -4 & 1 & & \\ & 1 & -4 & 6 & -4 & 1 & \\ & & 1 & -4 & 6 & -4 & 1 \\ & & & 1 & -4 & 6 & -4 \\ & & & & 1 & -4 & 5 \end{pmatrix}.$$

### 9.1 Program Text

**Note:** the listing of the example program presented below uses *bold italicised* terms to denote precision-dependent details. Please read the Users' Note for your implementation to check the interpretation of these terms. As explained in the Essential Introduction to this manual, the results produced may not be identical for all implementations.

```
*      F03ACF Example Program Text
*      Mark 14 Revised.  NAG Copyright 1989.
*      .. Parameters ..
INTEGER      NMAX, MMAX, IA, IL, M1MAX
PARAMETER   (NMAX=10,MMAX=4,IA=NMAX,IL=NMAX,M1MAX=MMAX+1)
INTEGER      NIN, NOUT
PARAMETER   (NIN=5,NOUT=6)
*      .. Local Scalars ..
real       DETERM
INTEGER      I, IFAIL, J, M, N
*      .. Local Arrays ..
real       A(IA,M1MAX), RL(IL,M1MAX)
*      .. External Subroutines ..
EXTERNAL    F03ACF
*      .. Executable Statements ..
WRITE (NOUT,*) 'F03ACF Example Program Results'
*      Skip heading in data file
READ (NIN,*)
READ (NIN,*) N, M
WRITE (NOUT,*)
IF (N.LT.1 .OR. N.GT.NMAX .OR. M.LT.1 .OR. M.GT.MMAX) THEN
  WRITE (NOUT,99999) 'N or M out of range: N = ', N, ' M = ', M
ELSE
  READ (NIN,*) ((A(I,J),J=1,M+1),I=1,N)
  IFAIL = 1
*
  CALL F03ACF(A,IA,N,M,DETERM,RL,IL,M+1,IFAIL)
*
  IF (IFAIL.NE.0) THEN
    WRITE (NOUT,99999) 'Error in F03ACF. IFAIL = ', IFAIL
  ELSE
    WRITE (NOUT,99998) 'Value of determinant = ', DETERM
  END IF
END IF
STOP
*
99999 FORMAT (1X,A,I5,A,I5)
99998 FORMAT (1X,A,F9.4)
END
```

## 9.2 Program Data

F03ACF Example Program Data

```
7 2
0 0 5
0 -4 6
1 -4 6
1 -4 6
1 -4 6
1 -4 6
1 -4 5
```

## 9.3 Program Results

F03ACF Example Program Results

Value of determinant = 64.0000

---